

# Processing 言語による情報メディア入門

## オブジェクト指向入門

神奈川工科大学情報メディア学科 佐藤尚

### はじめに

**最** 近のプログラミング言語では、オブジェクト指向 (object oriented) と呼ばれる機能を持っているものが多くあります。オブジェクト指向は、次の2つの仕組みを提供しようとするものです。

- 1) 複数のデータをまとめて一つに扱う仕組み。
- 2) 機能拡張を容易に行えるようにする仕組み。

そこで、円が上から下に移動するようなプログラムを考えてみます。これはサンプル 12-1 のようになります。どのように、操作するプログラムかはわかりますね。

### サンプルプログラム 12-1

```
float xBall;// 円の中心の X 座標
float yBall; // 円の中心の Y 座標
float rBall; // 円の半径
color cBall; // 円の色
void setup(){
  size(400,400);
  smooth();
  colorMode(HSB,359,99,99);
// 円の初期状態の決定
  rBall = random(10,20);
  xBall = random(rBall,width-rBall);
  yBall = -rBall;
  cBall = color(random(360),random(50,100),random(50,100));
}
void draw(){
  background(0,0,99);
// 円を移動させる
  yBall += 1;
  if(yBall-rBall > height){
    yBall = -rBall;
  }
// 円を描く
  stroke(cBall);
  fill(cBall);
  ellipse(xBall,yBall,2*rBall,2*rBall);
}
```

今度は1つの円ではなく、沢山の円を表示するようなサンプルを考えてみます。円の中心座標の値が入っている xBall や yBall などの値を配列変数にすることで、沢山の円に対する処理を簡単に記述できるようになります。サンプル 12-2 もどのような動作をしているかわかりますね。

### サンプルプログラム 12-2

```
int numberOfBalls = 100; // 円の個数
float[] xBall; // 円の中心の X 座標
float[] yBall; // 円の中心の Y 座標
float[] rBall; // 円の半径
color[] cBall; // 円の色
void setup(){
  size(400,400);
  smooth();
  smooth();
  colorMode(HSB,359,99,99);
  // 配列の確保
  xBall = new float[numberOfBalls];
  yBall = new float[numberOfBalls];
  rBall = new float[numberOfBalls];
  cBall = new color[numberOfBalls];
  // 円の初期状態の決定
  for(int i=0;i<numberOfBalls;i++){
    rBall[i] = random(10,20);
    xBall[i] = random(rBall[i],width-rBall[i]);
    yBall[i] = -rBall[i];
    cBall[i] = color(random(360),random(50,100),random(50,100));
  }
}
void draw(){
  background(0,0,99);
  for(int i=0;i<numberOfBalls;i++){
  // 円を移動させる
    yBall[i] += 1;
    if(yBall[i]-rBall[i] > height){
      yBall[i] = -rBall[i];
    }
  // 円を描く
    stroke(cBall[i]);
    fill(cBall[i]);
    ellipse(xBall[i],yBall[i],2*rBall[i],2*rBall[i]);
  }
}
```

サンプル 12-1 よりは少し複雑になっているように見えますが、沢山の円を処理するために、for 命令による繰り返し処理が付け加わっているだけです。そのために、xBall が xBall[i] などと置き換わっています。


サンプル 12-2 では、配列変数 xBall, yBall, rBall, cBall 毎に塊を作っています。しかし、プログラムでの意味的には、xBall[0], yBall[0], rBall[0], cBall[0] は、1つの円の中心座標、半径、色を保存しています。添え字の



番号の値が同じものどうして組を作って、プログラム内での意味を表しています。このサンプルのように、複数の変数が集まって、一つの意味のある情報を表すことがあります。Processing では、このような複数の情報をまとめて、新たなデータ型を作る仕組みが用意されています。それがクラスと呼ばれる仕組みです。最近のプログラミング言語は、この仕組みを持っていることが一般的になっています。

## データをまとめる仕組みとしてのクラス

簡単のために、**簡**サンプル 12-1 をクラスの仕組みを用いて書き直してみます。サンプル 12-1 でも、xBall, yBall, rBall, cBall が一塊となって、意味のある情報を表しています。どの情報かを区別するために、名前をつける必要があります。クラ



スを構成する個々の情報（データ）のことをメンバ (member) やメンバ変数と呼び、その名前をメンバ名と呼んでいます。そこで、xBall は円の中心の X 座標の値なので xCenter という名前で表すことにします。同様に、yBall は円の中心の Y 座標の値なので yCenter という名前で表すことにします。また、rBall は円の半径なので radius、cBall は円の色なので col とすることにします。また、この 4 つの情報を一塊にしたものを Ball と名付けることにします。この Ball はクラス名と呼ばれます。一般に、クラス名は大文字から始まる名前にし

ます。また、メンバ変数にはどのようなデータを記録するのかを指定するために、データ型を指定する必要があります。今回は、xCenter, yCenter, radius は float 型、color は color 型とします。つまり、PImage や PFont などはクラスという仕組みで作られたデータ型でした。

### クラスの宣言その1

クラスの宣言の一般形	Ball クラスの宣言
<pre>class クラス名 {   メンバ変数型0 メンバ名0;   メンバ変数型1 メンバ名1;   メンバ変数型2 メンバ名02   :   :   : }</pre>	<pre>class Ball {   float xCenter;   float yCenter;   float radius;   color col; }</pre>

この処理を行わないと、再度の実行の際に音声ファイルの再生などが正常に行われないなどの不都合が起きる場合があります。

このように定義したクラスは通常の変数と同じように利用することが出来ます。サンプル 12-1 をこの Ball クラスを使って書きかえたものがサンプル 12-3 です。

### サンプルプログラム 12-3

```
class Ball {
  float xCenter; // 円の中心の X 座標
  float yCenter; // 円の中心の Y 座標
  float radius; // 円の半径
  color col; // 円の色
}

Ball myBall; // Ball 型変数の宣言

void setup(){
  size(400,400);
  smooth();
  colorMode(HSB,359,99,99);
  // 円の初期状態の決定
  myBall = new Ball();
  myBall.radius = random(10,20);
  myBall.xCenter = random(myBall.radius,width- myBall.radius);
  myBall.yCenter = -myBall.radius;
  myBall.col = color(random(360),random(50,100),random(50,100));
}

void draw(){
  background(0,0,99);
  // 円を移動させる
  myBall.yCenter += 1;
  if(myBall.yCenter - myBall.radius > height){
    myBall.yCenter = -myBall.radius;
  }
}
```

```
// 円を描く
stroke(myBall.col);
fill(myBall.col);
ellipse(myBall.xCenter,myBall.yCenter,2*myBall.radius,2*myBall.
radius);
}
```

Ball クラスの変数を宣言するためには、通常の変数の宣言と同じように「Ball myBall;」などとします。また、実際にデータを保存する場所を作る必要があります。これを行っているのが、「myBall = new Ball();」の部分です。クラスはどのような種類のデータの集まりかを定める鋳型（テンプレート）のようなものです。この鋳型から new 関数を使って、実際にデータを保存する場所を作りだします。この作り出された場所のことをインスタンス (instance) と呼んでいます。鯛焼き器がクラスで、鯛焼きがインスタンス、鯛焼き器を使って鯛焼きを作る作業が new といった感じでしょうか？

姉ヶ崎寧々さんというキャラクタはクラスのようなもので、姉ヶ崎寧々さんは俺の嫁と思っている人の3DSにはインスタンスとしての“姉ヶ崎寧々さん”が存在しています。

インスタンスのメンバ変数にアクセスするためには、“.”を使います。例えば、myBall の xCenter にアクセスするためには、myBall.xCenter などとします。その他のメンバの値に対しても、同じようにアクセス出来ます。

サンプル 12-3 は、クラスを使ったプログラム例としては少し不自然なものです。実は、メンバには単なる変数だけでなく、関数を持つてくることも出来ます。クラスに付随している関数のことは、メソッド (method) と呼びます。メソッドの定義は、通常関数の定義と同じです。一つ異なっている点は、class クラス名 { ~ } の中に書くことになっている点です。また、メンバ変数の初期化などはコンストラクタ (constructor) と呼ばれる特殊なメソッド（戻り値無し、名前はクラス名と同じ）を利用します。

### クラスの宣言その2

クラスの宣言の一般形	Ball クラスの宣言
<pre>class クラス名 {   メンバ変数型0 メンバ名0;   メンバ変数型1 メンバ名1;   メンバ変数型2 メンバ名02   :   :   クラス名0{   ~   } }</pre>	<pre>class Ball {   float xCenter;   float yCenter;   float radius;   color col;   Ball(){   ~   } }</pre>



コンストラクタを使ってサンプル 12-3 を書きかえたものがサンプ

ル 12-4 です。この例では、class Ball { ~ } 内の Ball() { ~ } の部分が  
コンストラクタです。コンストラクタやそのメソッドが付随してい  
る class クラス名 { ~ } の部分でメソッドの定義を書く場合には、直  
接メンバ名を書けば、メンバ変数にアクセスすることが出来ます。

#### サンプルプログラム 12-4

```
class Ball {
  float xCenter; // 円の中心の X 座標
  float yCenter; // 円の中心の Y 座標
  float radius; // 円の半径
  color col; // 円の色
  Ball(){
    radius = random(10,20);
    xCenter = random(radius,width- radius);
    yCenter = -radius;
    col = color(random(360),random(50,100),random(50,100));
  }
}

Ball myBall; // Ball 型変数の宣言

void setup(){
  size(400,400);
  smooth();
  colorMode(HSB,359,99,99);
  // 円の初期状態の決定
  myBall = new Ball();
}

void draw(){
  background(0,0,99);
  // 円を移動させる ,update
  myBall.yCenter += 1;
  if(myBall.yCenter - myBall.radius > height){
    myBall.yCenter = -myBall.radius;
  }
  // 円を描く ,draw
  stroke(myBall.col);
  fill(myBall.col);
  ellipse(myBall.xCenter,myBall.yCenter,2*myBall.radius,2*myBall.
radius);
}
```

サンプル 12-4 の「円を移動させる」や「円を描く」などの部分は、  
一つのインスタンスだけの情報を利用して作られています。このよ  
うな場合には、クラスのメソッドとして書くことが一般的です。そ  
こで、このような方針でサンプル 12-4 を書きかえたものがサンプル  
12-5 です。クラスに付随するメソッドを呼び出す場合にも、メンバ  
変数と同じように "." を使って使用します。

### クラスの宣言その3

クラスの宣言の一般形	Ball クラスの宣言
<pre>class クラス名 {   メンバ変数型0 メンバ名0;   メンバ変数型1 メンバ名1;   メンバ変数型2 メンバ名02       :       :       :   クラス名0{       ~   }   戻り値型0 メソッド名0(仮引数の並び){       ~   }   戻り値型1 メソッド名1(仮引数の並び){       ~   }       : }</pre>	<pre>class Ball {   float xCenter;   float yCenter;   float radius;   color col;   Ball(){       ~   }   void update(){       ~   }   void draw(){       ~   } }</pre>

### サンプルプログラム 12-5

```
class Ball {
  float xCenter; // 円の中心のX座標
  float yCenter; // 円の中心のY座標
  float radius; // 円の半径
  color col; // 円の色
  // コンストラクタの定義
  Ball(){
    radius = random(10,20);
    xCenter = random(radius,width- radius);
    yCenter = -radius;
    col = color(random(360),random(50,100),random(50,100));
  }
  // メソッドの定義
  void update(){
    yCenter += 1;
    if(yCenter - radius > height){
      yCenter = -radius;
    }
  }
  void draw(){
    stroke(col);
    fill(col);
    ellipse(xCenter,yCenter,2*radius,2*radius);
  }
}
```

```

Ball myBall; // Ball 型変数の宣言
void setup(){
  size(400,400);
  smooth();
  colorMode(HSB,359,99,99);
// 円の初期状態の決定
  myBall = new Ball();
}
void draw(){
  background(0,0,99);
// 円を移動させる ,update
  myBall.update(); // myBall の update メソッドの呼び出し
// 円を描く ,draw
  myBall.draw(); // myBall の draw メソッドの呼び出し
}

```

今までのクラスを使ったサンプルでは、一つのタブの中に全てのプログラムを書いていた。しかし、通常はクラス毎に別々のタブに記述します。新たにタブを作るためには、次の様に行います。

1. ウィンドウの右上にある矢印状のボタンを押します。
2. すると、メニューが出てきますので、「New Tab」を選択します。
3. そして、新たに作るタブの名前を入力し、OK ボタンを押します。
4. 新しいタブが作られます。
5. タブの名前は、クラスの名前と同じにするのが一般的です。

Ball クラスを使って、サンプル 12-2 を書きかえてみます。この結果がサンプル 12-6 です。

### サンプルプログラム 12-6

#### Ball クラスのタブ

```

class Ball {
  float xCenter; // 円の中心の X 座標
  float yCenter; // 円の中心の Y 座標
  float radius; // 円の半径
  color col; // 円の色
// コンストラクタの定義
  Ball(){
    radius = random(10,20);
    xCenter = random(radius,width- radius);
    yCenter = -radius;
    col = color(random(360),random(50,100),random(50,100));
  }
// メソッドの定義
  void update(){
    yCenter += 1;
    if(yCenter - radius > height){
      yCenter = -radius;
    }
  }
}

```



```

void draw(){
    stroke(col);
    fill(col);
    ellipse(xCenter,yCenter,2*radius,2*radius);
}
}

```

メインのタブ

```

int numberOfBalls=100;
Ball[] myBalls; // Ball 型変数の宣言
void setup(){
    size(400,400);
    smooth();
    colorMode(HSB,359,99,99);
// 配列の確保
myBalls = new Ball[numberOfBalls];
for(int i=0;i<numberOfBalls;i++){
    myBalls[i] = new Ball();
}
}
void draw(){
    background(0,0,99);
    for(int i=0;i<numberOfBalls;i++){
        myBalls[i].update(); // myBalls[i] の update メソッドの呼び出し
        myBalls[i].draw(); // myBall[i] の draw メソッドの呼び出し
    }
}
}


```

このようにクラスを利用してプログラムを作成すると、プログラムの見通しが良くなります。また、コンストラクタにも引数を渡すことが出来ます。

#### クラスの宣言その4

クラスの宣言の一般形	Ball クラスの宣言
<pre> class クラス名 {     メンバ変数型 0 メンバ名 0;     メンバ変数型 1 メンバ名 1;     メンバ変数型 2 メンバ名 02         :         :     クラス名 () {         ~     }     戻り値型 0 メソッド名 0(仮引数の並び){         ~     }     戻り値型 1 メソッド名 1(仮引数の並び){         ~     }         : } </pre>	<pre> class Ball {     float xCenter;     float yCenter;     float radius;     color col;     Ball(){         ~     }     void update(){         ~     }     void draw(){         ~     } } </pre>

今まで、説明をサボってきましたが、クラス型の変数は、そのクラスのインスタンスへの参照となっています。普通は気にしなくても大丈夫ですが、時々問題が起きることがあります。つまり、Ball 型変数同士の代入を行っても、変数が指しているインスタンスの情報そのものが複製される訳ではありません。インスタンスの情報もコピーするような代入を浅いコピー (shallow copy) と呼んでいます。メンバ変数の型が何らかのクラス型になっている場合には、単に参照がコピーされるだけです。Processing でも、浅いコピーを実現するための clone メソッドが用意されています。逆に、完全なコピーを作るような代入を深いコピー (deep copy) と呼ばれています。深いコピーを実現するためには、複製を作るために時間がかかるので、どうしても深いコピーを使いたいときだけ利用します。ちょっと難しい話なので、詳しくは触れません。



サンプル 12-6 をまねして、円の代わりに正方形が落ちてくるようなプログラムを作成してみます。サンプル 12-7 を見るとわかるように、クラスを使ってプログラムを作成しておく、どの部分を変更すれば良いかが見やすくなっていることがわかんと思います。

### サンプルプログラム 12-7

```

Square クラスのタブ
class Square {
  float xCenter; // 中心の X 座標
  float yCenter; // 中心の Y 座標
  float length; // 一辺の長さ
  color col; // 色
  // コンストラクタの定義
  Square(){
    length = random(10,20);
    xCenter = random(length/2,width- length/2);
    yCenter = -length/2;
    col = color(random(360),random(50,100),random(50,100));
  }
  // メソッドの定義
  void update(){
    yCenter += 1;
    if(yCenter - length/2 > height){
      yCenter = -length/2;
    }
  }
}

```

```

void draw(){
    rectMode(CENTER);
    stroke(col);
    fill(col);
    rect(xCenter,yCenter,length,length);
}
}

```

### メインのタブ

```

int numberOfSquares=100;
Square[] mySquares; // Square 型変数の宣言
void setup(){
    size(400,400);
    smooth();
    colorMode(HSB,359,99,99);
// 配列の確保
    mySquares = new Square[numberOfSquares];
    for(int i=0;i<numberOfSquares;i++){
        mySquares[i] = new Square();
    }
}
void draw(){
    background(0,0,99);
    for(int i=0;i<numberOfSquares;i++){
        mySquares[i].update(); // mySquares[i] の update メソッドの呼び出し
        mySquares[i].draw(); // mySquare[i] の draw メソッドの呼び出し
    }
}
}

```

## 機能拡張・継承

**オ**ブジェクト指向の提供する仕組みの2つめは、機能拡張に関するものです。これは、ポケモンを進化させるように、元になるクラスを進化させ、機能を拡張しようとするものです。この進化の際には、もとのクラスの特徴（メンバ変数やメソッド）は保持され、新しい特徴が付け加わったり、元の特徴が進化したりします。オブジェクト指向の言葉では、元になるクラスを親クラス（スーパークラス）、進化して出来た新しいクラスを子クラスと呼んでいます。進化させることを継承する (inherit) と呼びます。

サンプル 12-8 では、JitteringObject というクラスを定義しています。このクラスは赤色の点を表示するものです。update メソッドは、乱数で点の位置を変更します。

## サンプルプログラム 12-8

```
class JitteringObject {
  float xCenter;
  float yCenter;
  JitteringObject(float x0, float y0){
    xCenter = x0;
    yCenter = y0;
  }
  void update(){
    xCenter = constrain(xCenter+random(-1,1),0,width);
    yCenter = constrain(yCenter+random(-1,1),0,height);
  }
  void draw(){
    stroke(255,10,10);
    point(xCenter,yCenter);
  }
}
JitteringObject myPoint;
void setup(){
  size(400,400);
  smooth();

  myPoint = new JitteringObject(width/2,height/2);
}
void draw(){
  background(255);
  myPoint.update();
  myPoint.draw();
}
```

このサンプルプログラムを変更して、点の代わりに正方形を表示するようにするための、一つの方法として、サンプル 12-9 のように行う方法があります。

## サンプルプログラム 12-9

```
class JitteringRect {
  float xCenter;
  float yCenter;
  float len;
  JitteringRect(float x0, float y0, float l0){
    xCenter = x0;
    yCenter = y0;
    len = l0;
  }
  void update(){
    xCenter = constrain(xCenter+random(-1,1),0,width);
    yCenter = constrain(yCenter+random(-1,1),0,height);
  }
}
≈
```

```

void draw(){
    stroke(255,10,10);
    fill(255,10,10);
    rectMode(CENTER);
    rect(xCenter,yCenter,len,len);
}
}

JitteringRect myRect;
void setup(){
    size(400,400);
    smooth();

    myRect = new JitteringRect(width/2,height/2,10);
}
void draw(){
    background(255);
    myRect.update();
    myRect.draw();
}

```

サンプル 12-8 とサンプル 12-9 を見比べると共通する部分が多いことに気がつきます。例えば、サンプル 12-9 では、サンプル 12-8 のコンストラクタ (JitteringObject() と JitteringRect()) の部分と draw メソッドが変わっています。このようなときに利用するのが継承という仕組みです。この継承を使ったものがサンプル 12-10 です。

### サンプルプログラム 12-10

```

class JitteringObject {
    float xCenter;
    float yCenter;
    JitteringObject(float x0,float y0){
        xCenter = x0;
        yCenter = y0;
    }
    void update(){
        xCenter = constrain(xCenter+random(-1,1),0,width);
        yCenter = constrain(yCenter+random(-1,1),0,height);
    }
    void draw(){
        stroke(255,10,10);
        point(xCenter,yCenter);
    }
}

```

```

class JitteringRect extends JitteringObject{
  float len;
  JitteringRect(float x0,float y0,float l0){
    super(x0,y0);
    len = l0;
  }
  void draw(){
    stroke(255,10,10);
    fill(255,10,10);
    rectMode(CENTER);
    rect(xCenter,yCenter,len,len);
  }
}
JitteringRect myRect;
void setup(){
  size(400,400);
  smooth();
  myRect = new JitteringRect(width/2,height/2,10);
}
void draw(){
  background(255);
  myRect.update();
  myRect.draw();
}

```

「class JitteringRect extends JitteringObject」の部分で、どのクラスから進化をさせて、新しいクラスを作るかを指定します。この場合では、JitteringObject クラスから進化をさせて（拡張させて、extends）、新しいクラス JitteringRect を作っています。元のクラスで定義されているメンバ変数やメソッドはそのまま利用することが出来ます。また、draw のように、子クラスでメソッドを書き換えることも出来ます。また、super と親クラスのコンストラクタを呼び出すことが出来ます。

## なぜオブジェクト指向の話をしたのか

**オ**ブジェクト指向の話をきちんと理解するためには、もう少し色々な説明をする必要があります。また、自分でクラスを作る際には、色々知っているといいことが沢山あります。これらの話題は、2年生のプログラミング関連の授業で詳しく説明されます。ですので、わざわざオブジェクト指向の話をする必要はなかったかもしれませんが、しかし、簡単でもオブジェクト指向の話をしたのには、訳があります。

前回にやった音を発生させる授業では、Minim クラスを使っていました。Processing を利用したプログラミングでは、全てのプログラムを自分で作成することもあります。他の人が作ったプログラムを利用しながら、プログラムを作成することがあります。この他人が

作ったプログラムを利用する際に必要となるのが、オブジェクト指向の言葉です。そこで簡単ですが、オブジェクト指向の話をしました。きちんと説明しませんが、サンプル 12-11 は円柱を表示するプログラムです。

### サンプルプログラム 12-11

```
// size(幅,高さ,OPENGL)の際のおまじない
import processing.opengl.*;

void setup(){
// size(幅,高さ,OPENGL) や size(幅,高さ,P3D) とすると、
// 3次元での表示が可能となる
  size(400,400,OPENGL);
  smooth();
}

void draw() {
  background(0);
  directionalLight(255,255,255,1,0,-1);
  noStroke();
  translate(width/2, height/2, 0);
  beginShape(QUAD_STRIP);
  for (int a = 0; a <= 360; a += 20) {
    float x = 100 * cos(radians(a));
    float z = 100 * sin(radians(a));
    normal(x, 0, z);
    vertex(x, -100, z);
    normal(x, 0, z);
    vertex(x, 100, z);
  }
  endShape();
}
```

これは、単純に円柱を表示するだけのプログラムです。マウスを動かしても円柱は動きません。マウスの動きに合わせて円柱を動かすためには、自分でプログラムを作成するという方法もありますが、他の人の使ったプログラムを利用するという方法もあります。ここでは、PeasyCam というものを利用してみます。このライブラリの web ページは <http://mrfeinberg.com/peasyCam/> です。このライブラリを利用すると簡単にカメラの移動を行うことができますようになります。この変更を加えてものがサンプル 12-12 です。

### サンプルプログラム 12-12

```
import processing.opengl.*;
import peasy.test.*;
import peasy.org.apache.commons.math.*;
import peasy.*;
import peasy.org.apache.commons.math.geometry.*;

PeasyCam cam;
```

```

void setup(){
  size(400,400,OPENGL);
  smooth();
// Peasy の機能を使うと、マウスの動きによって物体の回転ができる
  cam = new PeasyCam(this, width/2, height/2, 0,400);
// Peasy でカメラが表示する範囲を指定
  cam.setMinimumDistance(50);
  cam.setMaximumDistance(500);
}

void draw() {
  background(0);
// 光源の設定
  directionalLight(255,255,255,1,0,-1);
  noStroke();
  translate(width/2, height/2, 0);
  beginShape(QUAD_STRIP);
  for (int a = 0; a <= 360; a += 20) {
    float x = 100 * cos(radians(a));
    float z = 100 * sin(radians(a));
    normal(x, 0, z);
    vertex(x, -100, z);
    normal(x, 0, z);
    vertex(x, 100, z);
  }
  endShape();
}

```

一般的に、外部の人の作ったライブラリのインストールは簡単です。ユーザーのスケッチフォルダの中に libraries というフォルダを作り、その中にコピーするだけです。スケッチブックフォルダの場所は、Processing エディタメニューの [File]-[Preferences] で表示される設定ダイアログの [Sketchbook location:] に指定されているフォルダです。なお、コピーした後に、Processing のプログラムを再起動して下さい。

サンプル 12-12 の場合には、<http://mrfeinberg.com/peasycam/> の download から zip ファイルをダウンロード（2015 年 7 月 9 日時点では、peasycam\_201.zip）を解凍して得られるフォルダ peasycam をユーザーのスケッチフォルダの中に libraries にコピーするだけです。Proxy を使っていない環境であれば、Processing の機能を利用して (Sketch > Import Library... > Add Library...) でもインストールすることができます。なお、3 次元の物体の表示には、物体の位置や大きさ、光源の位置、視点の位置などさまざまな要素が絡み合うので、ちょっと複雑です。このサンプルでは、OBJ 形式と呼ばれている 3 次元形状の情報を保存するデータ形式のファイルから、表示する形状を読み込んでいます。



## サンプルプログラム 12-13

```
import peasy.test.*;
import peasy.org.apache.commons.math.*;
import peasy.*;
import peasy.org.apache.commons.math.geometry.*;
import processing.opengl.*;

PShape model;
PeasyCam cam;

void setup() {
  size(400, 400, OPENGL);
  // tachikoma.obj というファイルを読み込む
  model = loadShape("tachikoma.obj");
  // マウスの移動で表示物体を回転させる
  cam = new PeasyCam(this, width/2, height/2, 0,400);
  cam.setMinimumDistance(50);
  cam.setMaximumDistance(500);
}

void draw() {
  background(0, 0, 100);
  lights();
  translate(width/2, height/2, 0);
  // 0.5 という数字を変更すると、表示されている物体の大きさが変わる
  scale(0.5);
  shape(model,0,0);
}
```

## ファイルからのデータの読み出し

Processing 言語では、ファイルからのデータの読み出しや、書き込みを行うことができます。Processing 言語のバージョンが 2.2 以上の場合には、ファイルからのデータの読み出しや書き出しを行う関数が含まれています。しかし、バージョンが 1.5 の場合には、それほど多くはありません。ここでは、どちらでも使える、基本的な関数を紹介します。

ファイルからのデータの読み出しには、

- 1) 一気に行毎に全てのデータを読み出す、
- 2) 一行ずつデータを読み出す、
- 3) 一文字ずつデータを読み出す、

の 3 通りの方法があります。

一番簡単な方法は、1) の方法です。つぎのサンプルは "test.txt" というファイルからデータを読み出し、それを表示するプログラムです。

## サンプルプログラム 12-14

```
String[] moji;
PFont font;

void setup(){
  size(400,200);
  font = loadFont("Serif-48.vlw");
  textFont(font,36);
  // text.txt からデータを読み出す
  moji = loadStrings("text.txt");
}
void draw(){
  background(255);
  fill(0);
  textAlign(CENTER,CENTER);
  text(moji[second() % moji.length],0,0,width,height);
}
```

### text.txt の内容

Kongo	Fuso	Nagato
Hiei	Yamashiro	Mutsu
Haruna	Ise	Yamato
Kirishima	Hyuga	Musashi

ファイルからデータを読み出すために、loadStrings 関数を使用しています。この関数は、ファイルからデータを読み出し、そのデータを String 型の配列に保存するような関数です。このサンプルでは、text.txt というファイルからデータを読み出し、このファイルのデータを moji という String 型の配列に保存します。moji[0] には、ファイルの 1 行目のデータが文字列として記録されています。同様に、moji[1] にはファイルの 2 行目、moji[2] にはファイルの 3 行目の情報が記録されています。ファイルの最後の行のデータは moji[moji.length-1] に記録されています。loadStrings 関数では、ファイルのデータを一気に読み出し、メモリ中に保存するので、余り大きなファイルを読み込むことは難しいと思います。

### String[] loadStrings (ファイル名)

ファイル名で指定されたファイルからデータを読み出し、その結果を String 型の配列変数に保存する。読み出された情報は String 型となっています。また、一般的には、"Show sketch folder" で表示されるフォルダ内にある data フォルダに保存されているファイルからデータを読み出します。つまり、読み出したファイルはこの data フォルダに保存しておくことが必要となります。

loadStrings 関数を使った方法では、ファイルから読み出したデータはすべて文字列 (String) となってしまいます。数値データが入って

る場合には、どのようにしたら良いでしょうか？一番単純な方法は、文字列として読み込んだデータを int 型や float 型に変換する方法です。

次のサンプルでは、loadStrings 関数を使って fuel.txt というファイルからデータを読み出し、そのデータを int 関数によって、int 型の値に変換しています。

### サンプルプログラム 12-15

```
String[] moji;
int[] fuel;
int barWidth = 20;
void setup(){
  moji = loadStrings("fuel.txt");
  fuel = new int[moji.length];
  for(int i=0;i < moji.length;i++){
    fuel[i] = int(moji[i]);
  }
  size(moji.length*barWidth,max(fuel));
}
void draw(){
  background(255);
  stroke(255);
  for(int i=0;i<fuel.length;i++){
    fill(255-fuel[i]);
    rect(barWidth*i,height-fuel[i],barWidth,fuel[i]);
  }
}
```

### fuel.txt の内容

80	85	100
80	85	100
80	85	250
80	85	250

このサンプルでは、メモ帳などで読み込みに使うデータを作っていました。ゲームなどを作る際には、Excel などを利用してデータファイルを作成することがあります。拡張子が ".xlsx" などのファイルを読み込むことは面倒ですが、拡張子が ".csv" というファイルでは、割と簡単にデータを読み出すことができます。

次のような表の情報を保存した ".csv" ファイル ("battleships.csv") では、1 行の情報が "," で区切られて並んでいます。例えば、このファイル (battleships.csv) の情報を loadStrings 関数で読み込む (lines = loadStrings("battleships.csv");) とすると、lines[0] は "name,fuel,durableness" という文字列が記録されることとなります。同様に、lines[1] は "Kongo,80,63"、lines[2] は "Hiei,80,63"、

lines[3] は “Haruna,80,63”、lines[12] は “Musashi,250,94” という文字列が記録されることとなります。配列 lines に記録されている情報は、名前, 燃料, 耐久という情報が並んでいます。つまり、”,” で区切られている一つ一つの項目に意味があります。そこで、この意味のある情報をバラバラにして取り出すことが必要となります。このような目的に使える関数として split 関数があります。この関数は文字列を指定した文字を区切り文字を基準に分割する関数です。

### excel の表情報

name	fuel	durableness
Kongo	80	63
Hiei	80	63
Haruna	80	63
Kirishima	80	63
Fuso	85	67
Yamashiro	85	67
Ise	85	74
Hyuga	85	74
Nagato	100	80
Mutsu	100	80
Yamato	250	93
Musashi	250	94

### CVS ファイルの中身

name,fuel,durableness
Kongo,80,63
Hiei,80,63
Haruna,80,63
Kirishima,80,63
Fuso,85,67
Yamashiro,85,67
Ise,85,74
Hyuga,85,74
Nagato,100,80
Mutsu,100,80
Yamato,250,93
Musashi,250,94

### String[] split( 文字列, 区切り文字 )

文字列を区切り文字で分割し、その結果を String 型の配列に保存する関数です。

例えば、String[] items = split(“Kongo,80,63”,”,”) とすると、“Kongo,80,63” という文字列を区切り文字 “,” で分割するので、items[0] には “Kongo”、items[1] には “80”、items[2] には “63” という文字列が保存されることとなります。この利用したサンプルを次に示します。

## サンプルプログラム 12-16

```
PFont font;
String[] name;int[] fuel,durable;
int yourID;
int myID;
boolean playing = false;
void setup(){
  size(300,210);
  String[] lines = loadStrings("battleships.csv");
  font = loadFont("Serif-48.vlw");
  textFont(font,24);
  name = new String[lines.length-1];
  fuel = new int[lines.length-1];
  durable = new int[lines.length-1];
  for(int i=0;i<lines.length-1;i++){
    String[] items = split(lines[i+1],",");
    println(items);
    name[i] = items[0];
    fuel[i] = int(items[1]);
    durable[i] = int(items[2]);
  }
}
void mouseClicked(){
  yourID = int(random(name.length));
  myID = int(random(name.length));
  playing = true;
}
void draw(){background(255);
  fill(0);
  if(playing){
    textAlign(CENTER,CENTER);
    text("You choose "+name[yourID]+".",0,0,width,70);
    text("I choose "+name[myID]+".",0,70,width,70);
    String msg;
    if(durable[myID] > durable[yourID]){
      msg = "I win!!";
    }else if(durable[myID] == durable[yourID]){
      msg = "Draw!!";
    }else{
      msg = "You win!!";
    }
    text(msg,0,140,width,70);
  }
}
```

このように loadStrings 関数などを使うとファイルからデータを読み出すことが出来るようになります。

## ファイルへの保存

ファイルからのデータの書き出しには、読み出しの時と同じように、

- 1) 一気に毎行に全てのデータを書き出す、
- 2) 一行ずつデータを書き出す、
- 3) 一文字ずつデータを読み出す

の3通りの方法があります。一番簡単の方法は、1)の方法です。反対に、一番面倒な方法は3)の方法です。2)の1行ずつ書き出すという処理を行うことが多いように思います。ファイルに一気にデータ (String 型) を書き出すためには、saveStrings 関数を利用します。

### サンプルプログラム 12-17

```
String fruits = "apple banana orange strawberry";
String[] items = split(fruits, " ");
saveStrings("fruits.txt", items);
```

1行ずつ書き出すためには、最初に書き出すファイルを指定し、PrinterWriter と呼ばれるデータ型の変数を用意します。この変数から println メソッドなどを利用して、データを書き出して行きます。本来であれば、エラー発生に備えた処理を記述する必要があります。

### サンプルプログラム 12-18

```
PrintWriter output;
void setup(){
    size(400,400);
    output = createWriter("pos.txt");
}
void draw(){
    background(255);
    fill(255,0,0);
    ellipse(mouseX,mouseY,10,10);
    output.println(mouseX+", "+mouseY);
}
void mouseClicked(){
    // マウスをクリックすると、書き出し終了
    output.flush(); // メモリに残っている内容を書き出す
    output.close(); // 書き出し終了
    exit();
}
```

ファイルへの書き出しは、駆け足でごめんなさい。